
4 Modellierung elektrischer Netzwerke 

4.7 Komplexe Wechselstromrechnung und Ortskurven 

4.7.1 Rechnen mit komplexen Größen 
In den vorangegangenen Abschnitten des Kapitels 4 stand zwar die Berechnung von Aus-
gleichsvorgängen nach einer Änderung des Zustands von elektrischen Netzwerken im Vorder-
grund, die beschriebenen Rechnungen können aber auch Aussagen über die Netzwerkgrößen 
im stationären Zustand nach Ablauf des Einschwingvorgangs liefern, und zwar für beliebige 
Erregungen. Oft interessiert jedoch der Einschalt- oder Übergangsvorgang nicht, sondern nur 
der darauf folgende stationäre Zustand bei sinusförmigen Erregungen unter Beschränkung auf 
lineare Netzwerke. Für solche Fälle ist das Rechnen mit komplexen Größen - auch als symboli-
sche Methode bezeichnet - ein sehr leistungsfähiges Verfahren. Bei dieser Art der Rechnung 
werden die sinusförmigen Wechselgrößen in den ohmschen Widerständen, Induktivitäten  und 
Kapazitäten in entsprechende komplexe Größen abgebildet. Dadurch entstehen komplexe Ope-
ratoren, mit deren Hilfe komplexe Zeitfunktionen (rotierende Zeiger), komplexe Amplituden 
und komplexe Effektivwerte (ruhende Zeiger bei t = 0) ineinander überführt werden können 
[Weiß09]. Die für Gleichstromnetzwerke existierenden Gesetze (Abschnitt  4.1) gelten in ana-
loger Form für das Rechnen mit den komplexen Größen. 

Tabelle 4.3  Netzwerkelemente und komplexe Impedanzen 

Element Zeitbereich Bildbereich Z = u/ i 

Widerstand 
R 

( ) ( )u t R i t= ⋅  u R i= ⋅  RZ R=  

Induktivität 
L 

( ) ( )di t
u t L

dt
= ⋅  u j L iω= ⋅ ⋅  LZ j Lω= ⋅  

Kapazität 
C 

( ) ( )1du t
i t

dt C
= ⋅  

1j u i
C

ω ⋅ = ⋅  1
C

jZ
j C Cω ω

= = −
⋅ ⋅

 

Die Angaben in obiger Tabelle  haben Ähnlichkeit mit den Impedanz-Operatoren der Laplace-
Transformation (Tabelle 4.2). An Stelle der Variablen jω erscheint bei den Laplace-Trans-
formierten die Bildvariable s. Diese Ähnlichkeit darf aber nicht den unterschiedlichen Anwen-
dungsbereich beider Verfahren vergessen lassen. Den zwei Verfahren gemeinsam ist, dass sie 
nur auf  lineare Netzwerke anwendbar sind. Die mit der symbolischen Methode ermittelten 
Lösungen sind als Sonderfall in den mit der Laplace-Transformation ermittelten enthalten.  
Als bekannt vorausgesetzt werden im Folgenden die Grundlagen der komplexen Berechnung 
von Wechselstromschaltungen (siehe z. B. [Froh11, Weiß09]) und die Darlegungen des Kapi-
tels 2.6 zum Arbeiten mit komplexen Zahlen und Zeigerdarstellungen in Maple. Anhand von 
Beispielen wird demonstriert, wie man dieses Computeralgebra-System beim Rechnen mit 
komplexen Größen in der Wechselstromtechnik sowie bei der Darstellung der Ergebnisse nut-
zen kann.  

Komplexe Größen (Zeiger) werden in der Elektrotechnik heute meist durch einen Unterstrich 
gekennzeichnet. In den folgenden Maple-Programmen werden die entsprechenden Variablen 
für Spannungen und Ströme mit einem abschließenden Unterstrich versehen, wenn ohne diese 
Markierung eine Verwechslung mit dem Betrag der betreffenden Größe möglich wäre. Bei den 
komplexen Impedanzen wird auf diese Kennzeichnung verzichtet. 

Beispiel: Einfaches RC-Netzwerk 

Die Spannung u2(t) am Widerstand R2 und die komplexen Ströme in den Zweigen des Netz-
werks des Bildes 4.17 sollen berechnet werden.  

           
Bild  4.1  Einfaches RC-Netzwerk, 

im rechten Bild mit komplexen Größen 

Für die Eingangsspannung gilt 

( ) ( )2 cosu t U tω α= ⋅ ⋅ ⋅ +  

Bei der Lösung der vorliegenden Aufgabe soll mit Effektivwerten gerechnet werden. Die kom-
plexe Eingangsspannung ist somit 

jU U e α= ⋅  
> restart: with(plots):  
> interface(imaginaryunit = j): 

Komplexe Spannung U 
> G1:= U_ = U*exp(j*alpha); 

Gleichung für die Berechnung der Impedanz Zpar der Parallelschaltung: 
> G2:= Zpar = R2*1/(j*omega*C)/(R2+1/(j*omega*C)); 

 
Die Beziehung für die Spannung U2 am Widerstand R2 lässt sich sehr einfach mit Hilfe der 
Spannungsteilerregel formulieren.  
> G3:= U2_ = U_*Zpar/(R1+Zpar); 

Gleichung G2 wird nun in die Gleichung G3 eingesetzt. Dabei sind die folgenden Annahmen 
für Maple wichtig, um eine möglichst einfache Lösung zu finden. 
> assume(R1>0, R2>0, C>0, omega>0, alpha::real );  
> G4:= subs(G2, G3); 

 
> G4:= simplify(G4); 

 
Der Ausdruck für die komplexe Spannung U wird in Gleichung G4 eingesetzt: 
> G5:= subs(G1, G4); 

 
Nun wird die rechte Seite der Gleichung G5 der Variablen U2_ zugewiesen.  
> assign(G5); U2_; 

 
G5 wird für die folgende Rechnung nicht mehr benötigt. Es war daher akzeptabel, dass durch 
assign die linke Seite von G5 durch die rechte Seite überschrieben wurde.  
Die nächsten zwei Anweisungen bestimmen den Betrag (Absolutwert) und den Winkel des 
Zeigers U2_. 
> U2:= abs(U2_); 

 
> phi:= argument(U2_); 

 
Die komplexen Ströme in den Netzwerkzweigen liefern die folgenden Anweisungen: 
> I2_:= U2_/R2; IC_:= U2_*j*omega*C; I_:= I2_+IC_; 

> I_:= simplify(I_); 

 
Damit im Folgenden die Auswertung der gefundenen Beziehungen für unterschiedliche Para-
meterwerte möglich ist, werden diese den Variablen (z. B. R, C usw.) nicht fest zugewiesen, 
sondern in einer Parameterliste vereinbart. Mit dem Befehl subs werden anschließend die ge-
suchten Zahlenwerte berechnet.  
> param1:= R1=1, R2=2, C=0.001, omega=314, U=10, alpha=Pi/60: 

Komplexe Spannung am Widerstand R2: 
> subs(param1, U2_); 

 
Effektivwert U2: 
> subs(param1, U2); 

 
> simplify(%); 

 
Winkel von U2: Berechnung zuerst im Bogenmaß und nach Definition der Funktion grad im 
Gradmaß.  
> subs(param1, phi); 

 
> simplify(%); 

 
Definition der Funktion grad für die Umrechnung vom Bogen- ins Gradmaß: 
> grad:= arc -> simplify(180*arc/Pi); 

 
> grad(subs(param1, phi)); 

 
Aus den berechneten Werten folgt  

0,154
2 6,525 jU e− ⋅= ⋅   

bzw. nach Rücktransformation 

( ) ( )2 2 6,525 cos 8,8u t tω= ⋅ ⋅ ⋅ − °  

Für die komplexen Ströme in den Zweigen des Netzwerks ergeben sich folgende Werte: 
> simplify(subs(param1,I2_)); 

 
> simplify(subs(param1, IC_)); 

 
> simplify(subs(param1, I_)); 

 
Das Zeigerbild für die berechneten komplexen Größen soll nun dargestellt werden. Um den 
dafür erforderlichen Schreibaufwand zu verringern bzw. die Übersichtlichkeit zu erhöhen, wird 
die Prozedur zeiger definiert. 
> zeiger:= proc(x, ap, opt) 
    # Prozedur für einen Zeiger x mit dem Anfangspunkt ap  
    # opt...Optionen für Zeigerdarstellung (z.B. Farbe, Dicke) 
    local zeiger; 
    zeiger:= arrow([Re(ap),Im(ap)],[Re(x),Im(x)],op(opt)): 
  end proc: 

Dieser Prozedur sind die Optionen für die Zeigerdarstellung (opt) als Menge zu übergeben. Die 
darzustellende Größe wird über den formalen Parameter x übermittelt. Der Parameter ap legt 
den Anfangspunkt des Zeigers fest (siehe 2.6.). Die Prozedur gibt ein Plot-Konstrukt des Zei-
gers zurück. Dieses kann mit einem Befehl display dargestellt werden.  
Bei den folgenden Aufrufen der Prozedur zeiger wird die Übergabe der darzustellenden Größe 
mit der Vorgabe der Parameterwerte durch den Befehl subs verbunden. 
> Opt1:= {color=blue, width=0.1, head_length=0.5, head_width=0.3}: 
> zgI:= zeiger(subs(param1, I_), 0, Opt1): 
> zgI2:= zeiger(subs(param1,I2_), 0, Opt1): 
> zgIC:= zeiger(subs(param1,IC_), subs(param1,I2_), Opt1): 
> display(zgI,zgIC,zgI2,scaling=constrained, axes=none); 

 

Dieses Zeigerdiagramm der komplexen Ströme soll noch durch die Zeiger der Spannungen im 
Netzwerk und einige Bezeichnungen ergänzt werden. 
Die Spannung am Widerstand R1 ist die Differenz der komplexen Spannungen U und U2: 
> U1_ := U_-U2_;  

> U1_:= subs(G1, U1_); 

 
> U1_:= simplify(U1_); 

 
Erzeugung der Zeigerdarstellungen: 
> Opt2:= {color=gray, width=0.1, head_length=0.5, head_width=0.3}: 
> zgU:= zeiger(subs(param1,rhs(G1)), 0, Opt2): 
> zgU2:= zeiger(subs(param1,U2_), 0, Opt2): 
> zgU1:= zeiger(subs(param1,U1_), subs(param1,U2_), Opt2): 
> bez:= textplot([[1.9,1.3,"I"], [1.8,-0.8,"I2"], [6.0,0.8,"U"], 
                  [4.7,-1.2,"U2"]], font=[TIMES,14]): 
> display(zgI, zgIC, zgI2, zgU, zgU2, zgU1, bez, 
          scaling=constrained, axes=none); 

 

Die im Befehl textplot anzugebenden Positionen für die Darstellung der Texte kann man er-
mitteln, indem man die dargestellte Graphik markiert und den Mauszeiger an die gewünschte 
Stelle führt. Die Positionsangabe für diese Stelle wird dann in der Kontextleiste der Maple-
Nutzeroberfläche (siehe 2.1.1) angezeigt. 

4.7.2 Ortskurven 

Beispiel: Reihenschaltung R-L  
> restart: with(plots):  
> interface(imaginaryunit = j): 

Komplexe Impedanz Z: 
> Z:= R + j*omega*L; 

 
> R:= 2:  L:= 10^(-2): 
> Z;  

Darstellung der Ortskurve der komplexen Impedanz Z: 
> plot([Re(Z), Im(Z), omega=0..300], labels=[Re,Im],  
        caption="Ortskurve von Z", color=blue); 

 

Ermittlung der Ortskurve des komplexen Leitwerts Y 
> Y:= 1/Z; 

 
Y für ω = 200 Hz wird berechnet und zur Darstellung eines Zeigers (exakt: einer Linie) be-
nutzt: 
> Y200:= subs(omega=200, Y); 

 
> zeig:= plot([[0,0],[Re(Y200),Im(Y200)]], color=black): 
> pY:= plot([Re(Y), Im(Y), omega=0..infinity], 
             labels=[Re,Im],caption="Ortskurve von Y",color=blue):  
> bez:= textplot([0.26,-0.27, typeset(omega," = 200 Hz")]): 
> display(pY, zeig, bez, scaling=constrained); 

4.7.3 Komplexe Netzwerksberechnung mit dem Maschenstromverfahren 
Wegen der schon genannten Analogie zwischen der komplexen Rechnung und der Netzbe-
rechnung im Bildbereich der Laplace-Transformation kann hier das unter 4.5.2 beschriebene 
Berechnungsverfahren sinngemäß angewendet werden. Definiert man für jede Masche des 
vorgegebenen Netzwerks einen Maschenstrom, dann lässt sich für das Netzwerk gemäß Ab-
schnitt 4.5.2 eine Matrizengleichung aufstellen. 

⋅Z I = U     mit (4.16) 

Z … Matrix der komplexen Maschenimpedanzen 
I …  Spaltenmatrix der komplexen Maschenströme 
U … Spaltenmatrix der komplexen Quellenspannungen 

Gleichung (4.16), nach der Spaltenmatrix der Maschenströme umgestellt, ergibt 
1−= ⋅I Z U  (4.17) 

Über die Maschenströme können dann alle Zweigströme und Zweigspannungen berechnet 
werden. Ein einfaches Beispiel soll das Verfahren verdeutlichen.  

Beispiel: Netzwerk mit drei Maschen 

Für das in Bild 4.19 dargestellte Netzwerk mit der Spannung u(t) = √2·U1·cos(ω·t + α) sind die 
Maschenströme zu berechnen. 

 

Bild 4.19  Netzwerk mit drei Maschen 

Nach der Festlegung der Richtungen der Maschenströme (Bild 4.20) werden die komplexen 
Gesamtimpedanzen Z11, Z22 und Z33 der drei Maschen sowie die komplexen Koppelimpedan-
zen Zi,k abgelesen bzw. aufgeschrieben. 

Bild 4.20  Netzwerk wie Bild 4.19 mit komplexen Maschenströmen 

> restart:  
> interface(imaginaryunit = j): 

Komplexe Gesamtimpedanzen und Koppelimpedanzen: 
> Z11:= R1+R2+j*omega*L1: 
> Z22:= j*omega*L1+1/(j*omega*C1)+1/(j*omega*C2): 
> Z33:= R2+j*omega*L2+1/(j*omega*C2): 
> Z12:= -j*omega*L1: 
> Z13:= -R2: 
> Z23:= -1/(j*omega*C2): 

Weil die Impedanzmatrix symmetrisch ist und damit Zi,k = Zk,i gilt, kann mit den dargestellten 
Impedanzen die Matrix Z notiert werden: 
> Z:= Matrix([[Z11,Z12,Z13],[Z12,Z22,Z23],[Z13,Z23,Z33]]); 

 
Quellspannungsvektor und Maschenstromvektor: 
> U:= Vector([U1_,0,0]); 

> I:= Vector('[I1_,I2_,I3_]'); 

 
Inverse der Impedanzmatrix Z bestimmen: 
> with(LinearAlgebra): 
> Zinv:= MatrixInverse(Z): 

Effektivwertzeiger der Spannung u(t): 
> U1_:= U1*exp(j*alpha): 

Parameterwerte: 
> param:= R1=3, R2=3, L1=1, L2=0.5, C1=0.01, C2=0.01, U1=10,  
           alpha=45*Pi/180, omega=10: 

Komplexen Vektor der Maschenströme berechnen: 
> I:= Zinv.U: 

Komplexe Maschenströme I1, I2 und I3: 
> I1_:=subs(param,I[1]); I2_:=subs(param,I[2]); I3_:=subs(param,I[3]); 

 

 

 
Effektivwerte der Maschenströme: 
> I1:= abs(I1_); I2:= abs(I2_); I3:= abs(I3_); 

 
 
 

Vor der Berechnung der Winkel der komplexen Ströme wird die Funktion winkel definiert, die 
den Winkel einer komplexen Größe x im Gradmaß liefert. 
> winkel:= x -> evalf(180*argument(x)/Pi); 

 
> phi[1]:= winkel(I1_); phi[2]:= winkel(I2_); phi[3]:= winkel(I3_); 

Damit können die gesuchten Maschenströme im Zeitbereich angegeben werden:   

i1(t) ≈ √2·1,86*cos(10/s*t + 30,7°);    i2(t) ≈ √2·1,6*cos(10/s*t + 61,7°);   
i3(t) ≈ √2·3,33*cos(10/s*t + 45°) 

